EXAMPLE 4. This program defines the max hold function to User math A, defines a function to get the -3 dB Bandwidth as User Defined Function 1, and defines a signal tracking function as Sweep End Function A.

Program Listing:

```
 10 !DEFINE USER MATH
 20 DMA=MAX(MA,A)
 30 PRMA"MAX"
 40 UNITA"DB"
 50 !
 60 !DEFINE UDF
 70 DF1"MCF4;DELT1;WIDTH1;MKMX;DLCURS=-3"
 80 LBL1"-3DB BAND WIDTH"
 90 !
100 !DEFINE SEF
110 DFA"MKMX;MKCTR"
120 LBLA"SIGNAL TRACK"
130 !
140 END
```

Line No.	Description
20	Define and assign the max hold function as User Math A
30	Name this function 'MAX'
40	Set the unit to 'DB'
70	Define User Defined Function 1 to get the -3 dB Bandwidth
80	Label the softkey for UDF 1 as '-3DB BANDWIDTH'
110	Define and assign the signal tracking function to Sweep End Function A
120	Label the softkey for SEF A as 'SIGNAL TRACK'

6-5. HP-IB

The 4195A can be used as a component in a high performance HP-IB system with other instruments, desktop computers, and minicomputers to form an automated measurement system. HP-IB is Hewlett-Packard's implementation of IEEE 488-1978, Digital Interface for Programmable Instrumentation.

6-5-1. 4195A's HP-IB CAPABILITY

Table 6-6 lists the 4195A's IEEE Standard 488-1978, HP-IB capabilities and functions. These functions provide the means for an instrument to receive, process and transmit, commands, data, and status over the HP-IB bus.

Table 6-6. HP-IB Interface Capability

Code	Function
SH 1	Complete Source Handshake capability
AH 1	Complete Acceptor Handshake capability
T 5	Basic Talker; serial poll; unaddressed if MLA; Talk-Only
L 4	Basic Listener; unaddressed if MTA; no Listen-Only
SR 1	Complete Service Request capability
RL 1	Complete Remote/Local capability
DC 1	Complete Device Clear capability
DT 1	Complete Device Trigger capability
C 0	No Controller capability
E 1	Drivers are open-collector

NOTE

The 4195A does not have parallel polling capability.

6-5-2. HP-IB DEFINITION

The 4195A can be defined as Addressable, or Talk-Only.

1. Addressable Mode

In the addressable mode, the 4195A is set to a Listener or a Talker, and the 4195A's operation is controlled by an external controller. The address is selectable between 0 and 30. The default address is 17.

To set the 4195A to the addressable mode, press the **MORE** key, **'HP-IB define'** softkey and the **'ADDRESSABLE'** softkey (or enter the "**HADM1**" command from the Keyboard Input Line).

The HP-IB address is set using the **'HP-IB address'** softkey (or the "**ADRS=**" command). The procedure for setting the HP-IB address is shown next.

1. Press the **MORE** key in the SPECIAL FUNCTION area, and the **'HP-IB define'** softkey.

2. Press the **'HP-IB address'** softkey. The current address will be displayed on the Keyboard Input Line, as follows.

   ADRS= 17

NOTE

The HP-IB address is stored in battery back up memory. If you have stored 20 as the HP-IB address, the displayed number will be 20 not 17.

3. Type the new address using the keys in the ENTRY area to change the address.

4. Press the **ENTER/EXECUTE** key.

2. Talk-Only Mode

The 4195A must be set to the Talk-only mode when it is connected to a Listen-Only device.

To set the 4195A to the Talk-Only mode, press the **MORE** key, **'HP-IB define'** softkey and the **'TALK only'** softkey (or enter the "**HADM2**" command from the Keyboard Input Line or from a User Program).

6-5-3. 4195A'S HP-IB COMMANDS

The 4195A HP-IB commands used to control the 4195A from an external controller are introduced here. These HP-IB commands are categorized as HP-IB Bus Commands, 4195A Query Commands, and 4195A Device Dependent Commands.

1. HP-IB Bus Commands

HP-IB bus commands have the same meaning in all HP-IB systems. The bus commands used by the 4195A are described here. HP BASIC statements are used in the description of the command examples. The three letter command abbreviations used in the IEEE 488-1978 nomenclature are shown in parentheses following each statement. Here, it assumed that the 4195A's HP-IB address is 717.

ABORT I/O (IFC):

ABORT I/O halts all bus activity and deselects the 4195A.

ABORT 7

CLEAR LOCKOUT/SET LOCAL:

CLEAR LOCKOUT/SET LOCAL releases devices on the bus from the lockout mode and returns them to local (front panel) control. The difference between **CLEAR LOCKOUT/SET LOCAL** and **LOCAL** is in the addressing method used.

LOCAL 7

DEVICE CLEAR (SDC or DCL):

This command is used with an address to clear a particular device (**SDC:** selected device clear) or used without an address (**DCL:** clears all devices). The 4195A initializes itself when it receives this command (but the memory is not cleared). It is good programming practice to perform initialization at the very start of the program.

CLEAR 7: clears all devices on port 7.

CLEAR 717: clears the instrument with address 17.

LOCAL (GTL):

LOCAL returns control of a listening device to front panel control.

LOCAL 717

LOCAL LOCKOUT (LLO):

LOCAL LOCKOUT disables the **LOCAL** key of all devices on the bus. After this command is sent you will be unable to operate the 4195A from the front panel. Execute the **LOCAL** command to undo **LOCAL LOCKOUT**.

LOCAL LOCKOUT 7

REMOTE:

REMOTE sets the 4195A to the remote mode. When this command is sent, the front panel with the exception of the **LCL** key will be disabled. If **LOCAL LOCKOUT** is asserted then the front panel **LCL** key will also be disabled.

REMOTE 7: sets all devices on port 7 to remote

REMOTE 717: sets the instrument with address 17 to remote.

SPOLL:

SPOLL is the SERIAL POLLING command used to place the status byte of the addressed instrument on the bus. The eight bits of the status byte can be masked off and read to determine the 4195A's operating state. See paragraph 6-5-7 for more information on the status byte.

SPOLL(717): the instrument with address 17 is serial polled.

SERVICE REQUEST:

The 4195A sends an **SRQ** (Service Request) control signal when it requires the controller to perform a task. **SRQ** can be thought of as an interrupt which informs the controller that information is ready to be transmitted, or that an error condition exists in the instrument. When the 4195A sends an **SRQ**, it also sets Bit 6 of the status byte. Bit 6 is the **RQS** (Request Service) bit, sometimes referred to as the "status bit" in connection with polling. When the 4195A is serially polled, it clears the **RQS** bit and the **SRQ** line, one of the five management control lines of the system interface. Any bit in the status byte can initiate an **SRQ**. The status byte may be masked by the user to determine which bits caused the 4195A to set the **SRQ** line. See paragraph 6-5-7, for more status byte information.

TRIGGER (GET):

This command may be sent to a selected listener on the HP-IB bus. The 4195A must be in the addressable mode, and the trigger mode must be set to the external trigger mode, before the trigger message is sent.

TRIGGER 7 : Trigger all devices on port 7

TRIGGER 717 : Trigger the instrument with address 17

NOTE

See the BASIC Interface Techniques manual supplied with the computer, for a full description of the HP-IB bus commands.

2. 4195A QUERY Commands

When a QUERY command is entered, data is output to the 4195A's output buffer. These commands can be entered using a User Defined Function, Sweep End Function, User Program, via HP-IB, and from the Keyboard Input Line.

STB?

Reads the status byte. When **STB?** is entered, the status byte will be read as a decimal number. If this command is entered via HP-IB when the status byte is **' 01011011 '**, you will read **' 91 '**.. Refer to paragraph 6-5-6, for the details of the status byte.

REV?

Reads the 4195A's firmware revision number. When **REV?** is entered, the revision date code will be output via HP-IB in the following format.

yyzz

When **REV?** is entered from the Keyboard Input Line, the revision number is displayed on the System Message Line, in the following format.

Rev x.xx yy zz

where **x.xx:** version number
yy: released date (year)
zz: released date (week)

ID?

ID? identifies a device connected on the HP-IB bus. When the 4195A receives this command, it outputs a message saying what it is. If **ID?** is entered from the Keyboard Input Line, the following message is displayed on the System Message Line.

HP4195A NETWORK/SPECTRUM_ANALYZER OPT000

ERR?

ERR? reads the error code of the existing error in the 4195A. When this command is entered, the error code will be read as a decimal number. If **ERR?** is entered from the Keyboard Input Line, the error code is displayed on the System Message Line. The error codes are listed in Appendix C.

DISP?

DISP? outputs the data displayed on the System Message Line by the ' **DISP**" " ' command.

CMT?

CMT? outputs the data displayed in the Comment Area by the ' **CMT**" " ' command.

(register)**?**

The (register)? commands; **R0?**, **MKR?**, **A?**, and etc., output the data in the selected register.

NOTE

When a **QUERY** command except for the "(register)?" command is entered, the data is output in the ASCII format. The output format of the data output by the "(register)?" command, is in the format selected by the "**FMT1**", "**FMT2**", or "**FMT3**" commands. The details of data output formats are described in paragraph 6-5-5.

3. 4195A Device Dependent Commands

The 4195A device dependent commands have meaning only to the 4195A and its functions. Device dependent commands are unique commands which corresponds to the softkeys or keys on an instruments front panel. The details of these commands are described in Sections 4 and 5. The 4195A device dependent commands are listed in Appendix E.

6-5-4. HOW TO OUTPUT REGISTER DATA

The 4195A offers three data output formats to transfer certain types of register data to the controller on the HP-IB bus. Each of the three data formats have a different data transfer rate. The 4195A is enabled to output register data, by receiving the Query Command (**?**) as follows.

Syntax:



Examples:

1. Single variable	2. Array variable
OUTPUT 717; " R1? "	**OUTPUT 717; " A? "**
ENTER 717; R1	**ENTER 717; A(*)**

6-5-5. DATA OUTPUT FORMATS

This paragraph describes the three 4195A data output formats; ASCII type, IEEE 64-bit binary type, and IEEE 32-bit binary type.

1. **ASCII Type: FMT1**

   ASCII type (FMT1) is the default data output format. When FMT1 is active (**FMT1** command is entered), the 4195A transfers data in the ASCII format. Register data is represented in the following ASCII format.

   a) Real Type (32-bit) Register Data

   This data output format is used for registers which hold 32-bit floating point numbers. The syntax and the registers which use this data type are as follows.

   SYNTAX:



S: Sign
N: Number
E: Exponential Sign

REGISTER:

A	B	C	D	E	F	G	H	I	J
RA	RB	RC	RD	RE	RF				
MA	MB								
MFOR	MFOI	MFSR	MFSI	MFLR	MFLI	MFTR	MFTI	MFIR	MFII
MROR	MROI	MRSR	MRSI	MRLR	MRLI	MRTR	MRTI	MRIR	MRII
ZSG	ZSB	ZOR	ZOX						
TFOR	TFOI	TFSR	TFSI	TFLR	TFLI				
TROR	TROI	TRSR	TRSI	TRLR	TRLI				
MKRA	MKRB	DMKRA	DMKRB	SMKRA	SMKRB	LCURS	DLCURS		
EQVR	EQVL	EQVCA	EQVCB						
REF	DIV	BTM							
NVAL	SMTHR	SMTHX	SMTHL	SMTHC	RLOSS	VSWR			
PER1	PER2	PET1	PET2	PEP1	PEP2				

b) Real Type (64-bit) Register Data

This format is used for the registers that hold 64-bit floating point numbers. (Leading zeros will be replaced by spaces.)

SYNTAX:



S: Sign
N: Number

REGISTER:

OSC1	OSC2				
START	STOP	STEP	CENTER	SPAN	
MANUAL	FREQ	BIAS	DFREQ	X	
MKR	DMKR	SMKR	LCURSL	LCURSR	WID

NOTE

When the oscillator level unit is V, the data of these registers is transmitted as Real type (32-bit).

SYNTAX:

S: Sign
N: Number
E: Exponential Sign

REGISTER:

Z	ST	RBW	QV	Rn (n= 0 to 99)

c) Integer Type Register Data

This format is used for the registers that hold 16-bit integer numbers. (Leading zeros will be replaced by spaces.)

SYNTAX:



S: Sign
N: Number

REGISTER:

NOP	ATR1	ATR2	ATT1	ATT2	ERR	PTN

2. IEEE 64-BIT Binary Type: FMT2

This type is the 64-bit floating point binary specified in the IEEE Standard 728-1982. This is the same data format used by HP Technical computers such as the HP 9000 Series 200/300 computers. The syntax diagram and the data format for FMT2, are shown below. The 4195A does not output un-normalized data and '-0'.

SYNTAX:



DATA FORMAT:

SEEEEEEEEEEEFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF

where	**S**: Sign bit of the fractional part	1 bit
	E: Exponent part	11 bits
	F: Fractional part	52 bits
	e: All bits of Exponent part	
	f: All bits of Fractional part	

Real number (RN) can be defined as follows.

1) when $0 < e < 11111111111$ (2047)

$$RN = (-1)^S \times 2^{(e-1023)} \times \{ 1 + f/(2^{52}) \}$$

2) when S = 0, e = 0 and f = 0

RN = 0 (zero)

Example:

a) If the sign bit (S) is 1, the number of exponent part (e) is 10000000000 (1024) and the number of fractional part (f) is 0, the real number (RN) is -2.

b) If S= 0, f= 10000000000000000000000000000000000000000000000000000 (2^{51}), and e= 01111111111 (1023), then RN= +1.5.

3. IEEE 32-BIT Binary Type: FMT3

This data type is the 32-bit floating point binary specified in the IEEE Standard 728-1982. This data type has the fastest data transfer rate. The syntax diagram and the data format for FMT3, are shown below. The 4195A does not output un-normalized data and **'-0'**.

SYNTAX:



DATA FORMAT:

SEEEEEEEEFFFFFFFFFFFFFFFFFFFFF

where	**S**: Sign bit of the fractional part	1 bit
	E: Exponent part	8 bits
	F: Fractional part	23 bits
	e: All bits of Exponent part	
	f: All bits of Fractional part	

Real number (RN) can be defined as follows.

1) when $0 < e < 11111111$ (255)

$$RN = (-1)^S \times 2^{(e-127)} \times \{ 1 + f/(2^{23}) \}$$

2) when S = 0, e = 0 and f = 0

RN = 0 (zero)

Example:

a) If the sign bit (S) is 0, the exponent part (e) is 01111111 (127) , and the fractional part (f) is 01000000000000000000000 (2^{21}), the real number (RN) is +1.25.

b) If S=1, e= 10000000 (128), and f= 10000000000000000000000 (2^{22}), then RN= -3.

6-5-6. DATA TRANSFER RATE

As previously stated, each data format has a different data transfer rate. Table 6-7 lists the typical data transfer rates when an array variable register consisting of 401 register elements is used.

Table 6-7. Data Transfer Rate

Code	Transfer Time with an HP 9000 series 300 computer	
	Using ENTER command	Using TRANSFER command
FMT1	700 msec	———
FMT2	120 msec	90 msec
FMT3	50 msec	40 msec

6-5-7. STATUS BYTE

A service request will be generated when any bit in the status byte is set. The status byte is an 8-bit status word that the 4195A places on the HP-IB bus when it is serially polled.

1. 4195A's Status Byte

The value of each bit indicates the status of an internal 4195A function. Bits are set to "1" and reset to "0". The bit assignments of the status byte are listed in Table 6-8.

Table 6-8. Status Byte Assignments

BIT	Value	Description
0(LSB)	0/1	**End of Measurement (EOM)**
1	0/1	**End of Sweep (EOS)**
2	0	Not used (always 0)
3	0/1	**End Status**
4	0/1	**Trigger too Fast**
5	0/1	**Error Status**
6	0/1	**RQS (Request Service)**
7(MSB)	0	Not used (always 0)

The definition of each bit is as follows:

BIT 0 (End of Measurement: EOM):

BIT 0 is set when a single point measurement (including compensation/ calibration measurement) is completed.

BIT 1 (End of Sweep: EOS):

This bit is set when a sweep measurement is completed (measurement of all points in the sweep range is complete). This includes compensation/ calibration measurements.

BIT 2 (Not used):

This bit is always 0 (zero).

BIT 3 (End Status):

This bit is set when the following operations are completed.

1. Copy
2. User Program (ASP) execution
3. Calibration/compensation

BIT 4 (Trigger too Fast):

This bit is set when the trigger command is sent under the following conditions.

1. The instrument is set to the **int** trigger mode.

2. The instrument is busy taking a measurement even in **ext** trigger mode.

BIT 5 (Error Status):

Bit 5 is set if any error is detected, including all HP-IB, hardware, and operation errors.

BIT 6 (Request Service: RQS):

This bit is set when the service request is granted, and is cleared when a serial-poll is performed. Bit 6 is non-maskable.

BIT 7 (Not used):

This bit is always 0 (zero).

NOTE

1. The status byte is cleared, reset to 0, when the 4195A receives the **CLS** command.

2. The status byte is cleared by the controller's serial polling, while BIT 6 (Request Service: RQS) of the status byte is set to 1.

3. The status byte can be read by sending an **STB?** query command. The **STB?** query command does not clear the status byte.

2. Masking the Status Byte

The "**RQS**" command is used to mask the status byte. The syntax of the "**RQS**" command is:

RQS = n (n= 0 to 255)

Where **n** is a decimal number corresponding to the mask bit pattern used to enable/disable bits of the status byte. For example, if **n** is equal to 34 (00100010), bits 1 and 5 are enabled, as follows.

RQS= 34 (00100010):

Bit No. of Status Byte	**MSB** 7	6	5	4	3	2	1	**LSB** 0
Bit Pattern for RQS command	0	0	1	0	0	0	1	0

0= disable
1= enable

In this example, when a bit in the status byte is set, in this case either bit 1 or 5, a service request is generated. The default value of **RQS** is 0 (00000000: all bits disabled), no service request is generated.

Bit 6 (RQS) is non-maskable, and bits 2 and 7 are always 0, so masking these bits has no meaning. In other words, masking the status byte should be performed on the lower 6 bits (except for bit 2). All masking combinations can be covered by using a mask pattern between 0 and 63 for the value of **n** in the command **RQS= n**.

6-5-8. SENDING CHARACTERS BY HP-IB

To output a character string to an external device connected to the HP-IB bus, use the 4195A Device Dependent "**SEND**" command. The syntax of this statement is:



By entering this command to the 4195A, the characters between the two ' " (double quotation)' marks are transmitted on the bus. The 4195A must be configured as a TALKER, and externally connected devices are configured as LISTENERS.

NOTE

The "SEND" statement can be used in a multi-statement.

6-5-9. HP-IB INTERFACE RESTRICTIONS

The following restrictions must be adhered to when using an HP-IB interface.

-- The total length of cable in one bus system must be less than or equal to two meters times the number of devices connected on the bus (the HP-IB controller counts as one device) and the total length of cable must not exceed 20 meters.

-- The maximum number of devices that may be connected on one bus system is 15.

-- There are no restrictions on how the cables are connected together. However, it is recommended that no more than four piggyback connectors be stacked together on one device. The resulting structure could exert enough force on the connector mounting to damage it.

For example, a system containing six devices can be connected together with cables that have a total length of less than or equal to 12 meters (six devices × 2m/device = 12 meters). The individual length of cable may be distributed in any manner desired as long as the total length does not exceed the allowed maximum. If more than ten devices are to be connected together, cables shorter than two meters must be used between some of the devices to keep the total cable length less than 20 meters.

Figure 6-3 shows the interconnection of a typical HP-IB system. The HP-IB connector is firmly fastened using two bolts to keep it from working loose during use. Figure 6-4 shows an HP-IB interface connector. The 4195A uses all of the available HP-IB lines; therefore, damage to any connector pin will adversely affect its HP-IB operation.



Figure 6-3. Typical HP-IB System Interconnection

CAUTION

THE 4195A CONTAINS METRIC THREADED HP-IB CABLE MOUNTING STUDS. THE METRIC VERSION OF THE HP 10833A, B, C, OR D HP-IB CABLE FASTENERS ARE DISTINGUISHED FROM THE ENGLISH VERSION BY COLOR. ENGLISH THREADED FASTENERS ARE SILVER; METRIC THREADED FASTENERS BLACK. DO NOT ATTEMPT TO MATE SILVER AND BLACK FASTENERS TO EACH OTHER. IF YOU DO, THE THREADS OF EITHER OR BOTH WILL BE DAMAGED.



PIN	LINE
1	DI01
2	DI02
3	DI03
4	DI04
13	DI05
14	DI06
15	DI07
16	DI08
5	EOI
17	REN
6	DAV
7	NRFD
8	NDAC
9	IFC
10	SRQ
11	ATN
12	SHIELD–CHASSIS GROUND
18	P/O TWISTED PAIR WITH PIN 6
19	P/O TWISTED PAIR WITH PIN 7
20	P/O TWISTED PAIR WITH PIN 8
21	P/O TWISTED PAIR WITH PIN 9
22	P/O TWISTED PAIR WITH PIN 10
23	P/O TWISTED PAIR WITH PIN 11
24	ISOLATED DIGITAL GROUND

THESE PINS ARE INTERNALLY GROUNDED

HP-IB Interconnect Cables

Part Number	Length
10833A	1 m (3.3 ft)
10833B	2 m (6.6 ft)
10833C	4 m (13.2 ft)
10833D	0.5 m (1.6 ft)

Figure 6-4. HP-IB interfacing

6-5-10. PROGRAM EXAMPLES

This paragraph introduces program examples for controlling the 4195A via HP-IB, with an HP 9000 Series 300 computer.

Example 1: Measurement Example using Network measurement function

Example 2: Data Transfer
- (1) Using ASCII format
- (2) Using Binary 64 bit format
- (3) Using Binary 32 bit format

Example 3: Hard Copy
- (1) Using Plot mode
- (2) Using Print mode
- (3) Using Dump mode

Example 4: Setting up a User Program

Example 5: Setting up a Programmed Points Table

NOTE

Before running the following programs, use the '**HP-IB address**' softkey to set the 4195A's HP-IB address to **17**.

Example 1: This program configures the system to measure the -3 dB Bandwidth of a 450 MHz Band-Pass Filter, and to print out the Insertion Loss, and the -3 dB Band Width.

Program Listing:

```
10    ! MEASUREMENT AND ANALYSIS EXAMPLE
20    !
30    Ads=717
40    REMOTE Ads
50    CLEAR Ads
60    !
70    Mask=2 ! Bit 1 enables SRQ interrupts.
80    Status_byte_rqs=2 ! Bit 1 enables End of sweep bit of 4195A.
90    !
100   !********** SET UP THE MEASUREMENT CONDITION **********
110   !
120   OUTPUT Ads;"FNC1"
130   OUTPUT Ads;"RST"
140   OUTPUT Ads;"RQS=";Status_byte_rqs
150   OUTPUT Ads;"CMT'';SWM2"
160   OUTPUT Ads;"PORT1;GPP1;DSP1"
170   OUTPUT Ads;"SWP1;CENTER=450MHZ;SPAN=20MHZ"
180   !
190   !********** TRIGGER **********
200   !
210   ON INTR 7 GOTO 250
220   OUTPUT Ads;"SWTRG"
230   ENABLE INTR 7;Mask
240   GOTO 240
250   OFF INTR 7
260   OUTPUT Ads;"CLS"
270   !
280   !********** ANALYSIS **********
290   !
300   OUTPUT Ads;"SCL2;AUTO;SCL1;AUTO"
310   OUTPUT Ads;"MCF4;MKMX;DLCURS=-3DB;DELT1;WIDTH1"
320   !
330   !********** INPUT DATA **********
340   !
350   OUTPUT Ads;"WID?"
360   ENTER Ads;Wid
370   OUTPUT Ads;"MKRA?"
380   ENTER Ads;Il
390   !
400   !********** DISPLAY THE DATA **********
410   !
420   PRINT "-3dB BAND WIDTH (Hz) =",Wid
430   PRINT "INSERTION LOSS (dB) =",Il
440   !
450   LOCAL Ads
460   END
```

Line Number	Description
70	**MASK** is used to enable the Service Request interrupts.
80	**Status_byte_rqs** is used to enable the 4195A's End of sweep bit.
120	Select the Network measurement function.
130	Initialize the 4195A for a Network measurement.
140	Enable bit 1 (**EOS**: End of Sweep bit) of the 4195A's status byte.
150	Clear the comment area, and select the Single sweep mode.
160	Select input port T1/R1, measurement parameter T/R(dB)-θ, and Rectangular X-A&B display format.
170	Set the measurement range (sweep range).
210 - 250	Trigger the sweep measurement, and wait until it is completed (until a service request from the 4195A is generated).
260	Clear the 4195A's status byte.
300	Auto scale the data displayed on the 4195A's screen.
310	Select the "o&LCRS mode", moves the **o** marker to a maximum point, and move the Line Cursor to the position 3 dB less than the o marker's position.
350 - 380	Store the -3 dB Bandwidth in variable *Wid*, and the insertion loss in variable *Il*.
420 - 430	Print the -3 dB Bandwidth and the Insertion Loss.

Example 2: The 4195A has three data output formats; FMT1, FMT2, and FMT3 (refer to paragraph 6-5-5). A program example will be given for each of these formats. In the FMT2 format (binary 64-bit data output), the measurement data is contained in the lower 8-bytes of the 12 data bytes transmitted by the 4195A. In the FMT3 format (binary 32-bit data output), the measurement data is contained in the lower 4-bytes of the 8 data bytes transmitted by the 4195A. The following programs use only the lower 8 or 4 data bytes.

(1) ASCII format (FMT1)

Program Listing:

```
10    ! DATA TRANSFER WHEN USING THE ASCII FORMAT (FMT1)
20    !
30    OPTION BASE 1
40    DIM A(401)
50    Ads=717
60    REMOTE Ads
70    !
80    OUTPUT Ads;"FMT1;A?"
90    ENTER Ads;A(*)
100   !
110   FOR I=1 TO 401
120   PRINT "A(";I;")=",A(I)
130   NEXT I
140   !
150   LOCAL Ads
160   END
```

Line Number	Description
80	Select the ASCII format, and transmit data in the A register through the output buffer of the 4195A.
90	Store the data sent from the 4195A, in variable A.
110 - 130	Print variable A.

(2) Binary 64-bit format (FMT2)

Program Listing:

```
10    ! DATA TRANSFER WHEN USING THE BINARY 64 BIT FORMAT (FMT2)
20    !
30    OPTION BASE 1
40    DIM Junk$[4]
50    REAL A(401)
60    !
70    ASSIGN @Ads TO 717;FORMAT ON
80    REMOTE @Ads
90    !
100   OUTPUT @Ads;"FMT2;A?"
110   ENTER @Ads USING "#,4A";Junk$
120   ASSIGN @Ads;FORMAT OFF
130   ENTER @Ads;A(*)
140   !
150   FOR I=1 TO 401
160   PRINT "A(";I;")=",A(I)
170   NEXT I
180   !
190   LOCAL @Ads
200   END
```

Line Number	Description
70	Set the I/O path between the controller and the 4195A with the **FORMAT ON** attribute, the 4195A can only receive data in the **ASCII** format.
100	Select the Binary 64-bit data output format, and output data in the A register through the 4195A's output buffer.
110	Store the upper 4-bytes of the data sent from the 4195A, in Junk$. This data is not measurement data, so it is not used.
120	Set the I/O path between the controller and the 4195A to the **FORMAT OFF** attribute, the binary 64-bit data format is the same data format used by HP 9000 series 300 computers.
130	Store the lower 8-bytes of data in variable A. The lower 8-bytes of data A are binary 64-bit data.
150 - 170	Print variable A.

(3) Binary 32-bit format (FMT3)

Program Listing:

```
10    !(3) DATA TRANSFER WHEN USING THE BINARY 32 BIT (FMT3)
20    !
30    OPTION BASE 1
40    INTEGER A(802),Upper,Lower,I
50    REAL Aa(401)
60    DIM Junk$[4]
70    ASSIGN @Ads TO 717;FORMAT ON
80    REMOTE @Ads
90    !
100   OUTPUT @Ads;"FMT3;A?"
110   ENTER @Ads USING "#,4A";Junk$
120   ASSIGN @Ads;FORMAT OFF
130   ENTER @Ads;A(*)
140   !
150   FOR I=1 TO 401
160     Upper=A(I*2-1)
170     Lower=A(I*2)
180     IF Upper=0 AND Lower=0 THEN
190       Aa(I)=0
200     ELSE
210       Exp=SHIFT(SHIFT(Upper,-1),8)
220       Tem=SHIFT(SHIFT(Upper,-9),9)
230       Low=Lower
240       IF Lower<0 THEN Low=65536+Lower
250       Man=Tem*2^16+Low
260       Aa(I)=DROUND(SGN(Upper)*(2^(Exp-127)+Man*2^(Exp-150)),6)
270     END IF
280     PRINT "A(";I;")=";Aa(I)
290   NEXT I
300   !
310   LOCAL @Ads
320   END
```

Line Number	Description
70	Set the I/O path between the controller and the 4195A with the **FORMAT ON** attribute, the 4195A can only receive data in the **ASCII** format.
100	Select the Binary 32-bit data output format, and move the data the A register to the 4195A's output buffer.
110	Store the upper 4-bytes of the data sent from the 4195A in Junk$. This data is not measurement data, so it is not used.
120	Set the I/O path between the controller and the 4195A to the **FORMAT OFF** attribute, this data is not in the **ASCII** format.

Line Number	Description
130	Store the lower 4-bytes sent from the 4195A in variable A (INTEGER). The lower 4 bytes are Binary 32-bit data. This data is entered every 2 bytes.
160 - 170	Store the upper 2-bytes of the binary 32-bit data in variable **Upper**, and the lower 2-bytes in variable **Lower**.
180 - 190	If **Upper**= 0 and **Lower**= 0, store 0 (zero) in variable **Aa**.
210 - 250	If **Upper**≠ 0 or **Lower**≠ 0, store the exponent part in **Upper** in variable **Exp**, the fractional part in **Upper** in variable **Tem**, and the fractional part in **Lower** in variable **Low**. Store the complete fractional part in variable **Man**.
260	Store the arranged data in variable **Aa**. The equation used to arrange the data, is described in paragraph 6-5-5.
280	Print variable **Aa**.

Example 3: The 4195A can plot, print, or dump the measurement data without an external controller (refer to paragraph 5-13). The following tells how to plot, print, and dump measurement data, via HP-IB.

(1) Plot (CPYM1)

Program Listing:

```
10    ! COPY DISPLAY BY "PLOT MODE" (CPYM1)
20    !
30    !********** INITIAL SETTING **********
40    !
50    INTEGER Select_code,Ads_4195a,Ads_plotter,Hp_4195a
60    Select_code=7
70    Ads_4195a=17
80    Ads_plotter=5
90    Hp_4195a=Select_code*100+Ads_4195a
100   !
110   Mask=2          ! Bit 1 enables SRQ interrupts.
120   Status_byte=8   ! Bit 3 enables End bit of 4195A.
130   !
140   ! (( PLOT AREA ))
150   !
160   P1x=2000        ! P1x is left of plot area
170   P1y=800         ! P1y is bottom of plot area
180   P2x=9200        ! P2x is right of plot area
190   P2y=7200        ! P2y is top of plot area
200   !               ( where 1 count is 0.025 mm )
210   !
220   REMOTE Hp_4195a
230   OUTPUT Hp_4195a;"RQS=";Status_byte
240   !
250   !********** PLOT GRATICULE **********
260   !
270   ON INTR Select_code GOTO End_plot
280   !
290   OUTPUT Hp_4195a;"CPYM1"
300   OUTPUT Hp_4195a;"PLTF1;SCLP1"
310   OUTPUT Hp_4195a;"PSCALE=";P1x;",";P1y;",";P2x;",";P2y
320   OUTPUT Hp_4195a;"SENDPS"
330   SEND Select_code;UNL TALK Ads_4195a LISTEN Ads_plotter DATA
340   WAIT .5
350   OUTPUT Hp_4195a;"COPY"
360   SEND Select_code;UNL TALK Ads_4195a LISTEN Ads_plotter DATA
370   !
380   ENABLE INTR Select_code;Mask
390   DISP "WAITING FOR PLOT"
400   GOTO 400
410   !
420 End_plot:  !
430   OFF INTR Select_code
440   OUTPUT Hp_4195a;"CLS"
450   DISP "PLOT COMPLETED"
460   END
```

Line Number	Description
70 - 90	Set the HP-IB addresses of the peripherals.
110	**Mask** is used to enable the Service Request interrupts.
120	**Status_byte** is used to mask the 4195A's status byte.
230	Bit 3 (END bit) of the 4195A's status byte is enabled.
290 - 320	Selects the 'plot mode', 'plot all', 'P1 P2 normal', and sets the plot area. Plot scale data is put in the 4195A's output buffer.
330	Configure the 4195A as a Talker, and the plotter as a Listener. Transmit plot scale data from the 4195A to the plotter.
340	Wait until the plot scale data is received by the plotter.
350	Send the "**COPY**" command to the 4195A. The 4195A outputs the data through its output buffer.
360	Configure the 4195A as a talker, and the plotter as a Listener. Transmit the data from the 4195A to the plotter.
380 - 400	Wait until the copy is completed (a service request from the 4195A is generated).
440	Clear the 4195A's status byte.

(2) Print (CPYM2)

Program Listing:

```
10    ! COPY MEASURED DATA BY "PRINT MODE" (CPYM2)
20    !
30    !********** INITIAL SETTING **********
40    !
50    INTEGER Select_code,Ads_4195a,Ads_prntr,Hp_4195a
60    Select_code=7
70    Ads_4195a=17
80    Ads_prntr=1
90    Hp_4195a=Select_code*100+Ads_4195a
100   !
110   Mask=2          ! Bit 1 enables SRQ interrupts.
120   Status_byte=8   ! Bit 3 enables End bit of 4195A.
130   !
140   REMOTE Hp_4195a
150   OUTPUT Hp_4195a;"RQS=";Status_byte
160   !
170   !********** PRINT DATA **********
180   !
190   ON INTR Select_code GOTO End_print
200   !
210   OUTPUT Hp_4195a;"CPYM2"
220   OUTPUT Hp_4195a;"COPY"
230   !
240   SEND Select_code;UNL TALK Ads_4195a LISTEN Ads_prntr DATA
250   ENABLE INTR Select_code;Mask
260   DISP "WAITING FOR PRINT"
270   GOTO 270
280   !
290 End_print:  !
300   OFF INTR Select_code
310   OUTPUT Hp_4195a;"CLS"
320   DISP "PRINT COMPLETED"
330   END
```

Line Number	Description
150	Enable bit 3 (END bit) of the 4195A's status byte.
210	Select the print mode.
220	Send the "**COPY**" command to the 4195A, the 4195A outputs the data through its output buffer.
240	Configure the 4195A as a Talker, and the printer as a Listener. Transmit the data from the 4195A to the printer.
250 - 270	Wait until the copy is completed (a service request from the 4195A is generated).
310	Clear the 4195A's status byte.

(3) Dump (CPYM3)

Program Listing:

```
10    ! COPY DISPLAY BY "DUMP MODE" (CPYM3)
20    !
30    !********** INITIAL SETTING **********
40    !
50    INTEGER Select_code,Ads_4195a,Ads_prntr,Hp_4195a
60    Select_code=7
70    Ads_4195a=17
80    Ads_prntr=1
90    Hp_4195a=Select_code*100+Ads_4195a
100   !
110   Mask=2          ! Bit 1 enables SRQ interrupts.
120   Status_byte=8   ! Bit 3 enables End bit of 4195A.
130   !
140   REMOTE Hp_4195a
150   OUTPUT Hp_4195a;"RQS=";Status_byte
160   !
170   !********** DUMP DISPLAY **********
180   !
190   ON INTR Select_code GOTO End_dump
200   !
210   OUTPUT Hp_4195a;"CPYM3"
220   OUTPUT Hp_4195a;"COPY"
230   !
240   SEND Select_code;UNL TALK Ads_4195a LISTEN Ads_prntr DATA
250   ENABLE INTR Select_code;Mask
260   DISP "WAITING FOR GRAPHICS DUMP"
270   GOTO 270
280   !
290 End_dump:  !
300   OFF INTR Select_code
310   OUTPUT Hp_4195a;"CLS"
320   DISP "GRAPHICS DUMP COMPLETED"
330   END
```

Line Number	Description
150	Enable bit 3 (END bit) of the 4195A's status byte.
210	Select the dump mode.
220	Send the "**COPY**" command to the 4195A, the 4195A puts the data in its output buffer.
240	Configure the 4195A as a Talker, and the printer as a Listener. Transmit the data from the 4195A to the printer.
250 - 270	Wait until the copy is completed (a service request from the 4195A is generated).
310	Clear the 4195A's status byte.

Example 4: This program sets up the User Program which is the User Program sample introduced in paragraph 6-4-9, Example 1.

Program Listing:

```
10    ! USER PROGRAM DOWNLOAD
20    !
30    Ads=717
40    REMOTE Ads
50    OUTPUT Ads;"SCRATCH"
60    !
70    OUTPUT Ads;"PROG""10 CMT'RIPPLE MEAS.'"""
80    OUTPUT Ads;"PROG""20 FNC1"""
90    OUTPUT Ads;"PROG""30 RST"""
100   OUTPUT Ads;"PROG""40 GPP1;PORT1"""
110   OUTPUT Ads;"PROG""50 CENTER=100MHZ;SPAN=500KHZ"""
120   OUTPUT Ads;"PROG""60 SWTRG"""
130   OUTPUT Ads;"PROG""70 MCF2;MKR=99990000;SMKR=100010000"""
140   OUTPUT Ads;"PROG""80 ARSTR;ANA1"""
150   OUTPUT Ads;"PROG""90 MKACT1;MKMX"""
160   OUTPUT Ads;"PROG""100 MKACT0;MKMN"""
170   OUTPUT Ads;"PROG""110 DELT1"""
180   OUTPUT Ads;"PROG""120 R1=DMKRA"""
190   OUTPUT Ads;"PROG""130 DISP'RIPPLE(DB)=',R1"""
200   OUTPUT Ads;"PROG""140 END"""
210   !
220   LOCAL Ads
230   BEEP
240   DISP "USER PROGRAM DOWNLOAD IS COMPLETE"
250   END
```

Line Number	Description
40	Set the 4195A to remote.
50	Clear the 4195A's ASP edit page.
70 - 200	Enter the User Program on the 4195A's ASP edit page, by using the "**PROG**" command.

Example 5: This program sets up a Programmed Points Table. The table set up in this example, is same as the table set up by the User Program in paragraph 6-4-9, Example 3.

Program Listing:

```
10    ! PROGRAMMED POINTS TABLE DOWNLOAD
20    !
30    Ads=717
40    REMOTE Ads
50    CLEAR Ads
60    !
70    OUTPUT Ads;"CPL1"
80    OUTPUT Ads;"PTN=1"
90    OUTPUT Ads;"PTCLR"
100   OUTPUT Ads;"PTSWP1"
110   !
120   Freq=190000000
130   !
140   FOR I=1 TO 401
150     OUTPUT Ads;"POINT=";Freq
160     IF I<101 OR I>300 THEN
170       Freq=Freq+80000
180       GOTO 220
190     ELSE
200       Freq=Freq+20000
210     END IF
220   NEXT I
230   !
240   LOCAL Ads
250   BEEP
260   DISP "PROGRAMMED POINTS TABLE DOWNLOAD IS COMPLETE"
270   END
```

Line Number	Description
70	Select the Couple mode to enter the RBW value to be coupled to the frequency (the Programmed Points Table must be set up with the sweep points, and the value of the Resolution Band Width).
80	Select Programmed Points Table number 1.
90	Clear table 1.
100	Set frequency sweep mode for Programmed Points Table 1.
150	Set the value indicated by **Freq** to the sweep point, by using the "**POINT=**" command.
160 - 180	If the number of points is less than 101, or greater than 300, add 80000 to **Freq**, and go to line 160.
190 - 210	If the number of points is between 101 and 300, add 20000 to **Freq**.

NOTES